



HAL
open science

Documentation du Programme 'ONDEL': LOGICIEL DE DECOMPOSITION DE SIGNAUX EN "ONDELETTES"

Ginette Saracco

► **To cite this version:**

Ginette Saracco. Documentation du Programme 'ONDEL': LOGICIEL DE DECOMPOSITION DE SIGNAUX EN "ONDELETTES". Note-Laboratoire de Mécanique & d'Acoustique, CNRS-UPR 7051, 1987, 5/87, pp.1-50. hal-00436546v2

HAL Id: hal-00436546

<https://hal.science/hal-00436546v2>

Submitted on 28 Feb 2011

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

LABORATOIRE DE MECANIQUE ET D'ACOUSTIQUE



DOCUMENTATION DU PROGRAMME 'ONDEL'.
LOGICIEL DE DECOMPOSITION DE SIGNAUX EN "ONDELETTES".

Ginette SARACCO

Note technique n° 5/87

25 JUILLET 1987



DOCUMENTATION DU PROGRAMME 'ONDEL'
LOGICIEL DE DECOMPOSITION DE SIGNAUX EN
ONDELETTES

par Ginette Saracco

25 Juillet 1987

DOCUMENTATION DU PROGRAMME ONDEL. LOGICIEL DE DECOMPOSITION DE SIGNAUX EN ONDELETTES.

SOMMAIRE

1. La transformée en ondelette
 - 1.1 Introduction générale
 - 1.2 Propriétés de l'ondelette analysante
 - 1.3 La transformée en ondelette
 2. Description du logiciel
 - 2.1 Caractéristiques générales
 - 2.2 Ondelette
 - 2.3 Grille
 - 2.4 Organigramme
 - 2.4.1 Présentation générale
 - 2.4.2 Organigramme
 - 2.4.3 Présentation des différents modules
 - a. Initialisation
 - i) Signal
 - ii) Ondelette
 - iii) Grille
 - b. Chargement du signal
 - c. Tests d'erreur
 - d. Transformée
 - e. Stockage des résultats de l'analyse
 3. Génération de signaux tests
 4. Exemple d'analyse sur les fonctions tests
 5. Programmes sources
- Annexe

DOCUMENTATION DU PROGRAMME ONDEL. LOGICIEL DE DECOMPOSITION DE SIGNAUX EN ONDELETTES.

1- La Transformée en Ondelettes.

1.1-Introduction générale.

La Transformée en Ondelettes est une méthode qui permet d'analyser et de resynthétiser un signal quelconque $s(p)$ en fonction de contributions élémentaires appelées ondelettes.

Toutes les ondelettes ont même forme et sont obtenues à partir de translations et dilatations (ou compressions) d'une fonction "mère" $g(p)$ (ondelette analysante). Elles constituent ainsi une famille à deux paramètres.

Chaque ondelette est limitée aussi bien dans le domaine fréquentiel que temporel. Cette transformation permet d'analyser un signal en conservant les caractéristiques de temps et de fréquence. L'énergie du signal est elle-même conservée.

Nous allons définir la notion d'ondelette dans le cas de signaux unidimensionnels, étant donné que tout pourra être étendu au cas de signaux multidimensionnels (p pouvant être le quadrivecteur espace-temps).

1.2-Propriétés de l'ondelette analysante.

Nous avons de vastes possibilités dans le choix de l'ondelette analysante $g(p)$. Cependant elle doit vérifier certaines conditions qui se résument ainsi :

- (i) $g(p)$ doit être régulière, et d'énergie finie.
- (ii) $g(p)$ est localisée à la fois en temps et en fréquence.
- (iii) g est à valeur moyenne nulle : $g^{\wedge}(0) = 0$ ou $\int g(p)dp = 0$

(g^{\wedge} est la transformée de Fourier de g)

La représentation graphique d'un exemple d'ondelette est donnée en annexe.

1.3-La transformée en Ondelettes.

On choisit une ondelette analysante $g(p)$. A partir de cette ondelette "mère", on construit une famille d'ondelettes élémentaires par dilatations et translations :


$$g^{(b,a)}(p) = g((p-b)/a) \quad ,$$

$a > 0$ est le paramètre de dilatation ($a < 1$ correspond à une compression),
 b quelconque est le paramètre de translation.

La transformée en Ondelette d'un signal $s(p)$ est obtenue en effectuant les produits scalaires du signal avec les ondelettes élémentaires :

$$S(b,a) = a^{-1/2} \int g^*((p-b)/a) s(p) dp \quad , \text{ où } g^* \text{ représente le complexe conjugué de } g.$$

$S(b,a)$ est une fonction définie sur le demi-plan ouvert :



Une propriété cruciale est la possibilité de reconstruire le signal à partir de sa transformée, par la formule:

$$s(p) = c^{ste} \iint S(b,a) g^{(b,a)}(p) da db / a^2$$

(La c^{ste} ne dépend que de g)

Tout signal apparait ainsi comme la superposition d'ondelettes élémentaires.

Sous certaines conditions peu exigeantes, il existe une formule discrète de reconstitution du signal, où n'est employée que la famille des ondelettes :

$$g(k2^{-j}, 2^{-j}), \quad j \text{ et } k \in \mathbb{Z}$$

2-Description du logiciel.

2.1-Caractéristiques générales.

- Le programme effectue la transformée en ondelettes d'un signal arbitraire réel ou entier stocké sur un fichier dont la longueur de l'enregistrement est paramétrée.
- La transformée est calculée pour 512 points du signal et stockée dans un fichier sous forme matricielle de N voies par 512 points ($N \leq 512$).
- Nous avons en sortie de la transformée deux cartes correspondant à l'analyse du signal donnant la répartition énergétique (module) et la phase instantanée (phase).
- Toutes les dimensions sont paramétrées de façon à ce que l'on puisse augmenter la taille des tableaux (signal, ondelette, transformée).

2.2-Ondelette.

- L'ondelette implémentée est une ondelette type Morlet. C'est une gaussienne modulée de la forme:

$$g(t) = G(t) e^{i\omega_0 t}, \quad \text{avec} \quad G(t) = e^{-t^2/2\sigma^2}$$

La transformée de Fourier étant : $g^{\wedge}(\omega) = G(\omega) \delta(\omega - \omega_0)$

où ω_0 représente la pulsation, σ l'écart type, nous avons donc une gaussienne décalée de $\omega_0/2\pi$ dans l'espace des fréquences.

Son support temporel est $t_{\max} - t_{\min}$ pour une valeur ϵ que nous nous fixons, valeur pour laquelle l'enveloppe de $G(t)$ est numériquement nulle.

- En résumé cette ondelette vérifie les conditions citées en 1.b. En outre la condition 1.b (iii) impose que ω_0 soit suffisamment grand de façon à approcher au mieux $\int g(t)dt = 0$. En pratique on prendra $\omega_0 \geq 5$.

- L'ondelette est discrétisée en NP points. Le nombre NP choisi correspond au paramètre de dilatation $a = 1$. Il nous permet de déduire la valeur du support de l'ondelette en fonction de la valeur ϵ . L'ondelette ainsi créée est échantillonnée à la fréquence d'échantillonnage f_e du signal.

- En annexe, nous pouvons voir la représentation graphique de cette ondelette pour les paramètres $a = 10$, $\omega_0 = 5$, $NP_{(a=1)} = 101$, $\epsilon = 10^{-2}$.

2.3-Grille.

La transformée définie par : $S(a,b) = a^{-1/2} \int g((t-b)/a) S(t) dt$

(où a,b représentent respectivement les paramètres de dilatation et translation), est calculée en discret sur une grille.

Le choix de cette grille détermine la nature des procédures de reconstitution du signal (formules d'interpolation), non implémentées ici, ainsi que l'aspect réduction de données (exemple grille dyadique).

Ici le choix de la grille est celui dite de la grille fine. C'est à dire que les paramètres seront définis par :

$$a = a_0 2^{k/NVO}$$

$$b = 1$$

(k numéro de la voie, NVO nombre de voies par octave)

2.4-Organigramme.

2.4.1 Présentation générale.

La procédure est initialisée à partir de la donnée de paramètres définissant la grille et l'ondelette. Nous avons ainsi une grande liberté dans le choix des caractéristiques de l'ondelette, du nombre de voies et de leur répartition. Ceci nous permet d'analyser le signal à la précision que l'on souhaite, dans les limites du principes d'incertitude d'Heisenberg.

Des messages (test d'erreurs) peuvent intervenir, à savoir :


- Risque d'aliasing (nombre de points de l'ondelette insuffisant pour le nombre de cycle de celle-ci, voir en annexe);

- Ondelette trop grande pour le paramètre de dilatation maximum demandé;

- Temps de départ de l'analyse trop petit. Dans ce cas nous avons la possibilité de continuer l'analyse en prolongeant le signal par des zéros. L'erreur en aval du signal se traduit par la présence d'une discontinuité localisée au début du fichier (voir §4. fig 4).

Cette présentation étant faite, nous allons décrire l'organigramme du programme.

2.4.2 Organigramme.



2.4.3 Presentation des différents modules.

a. INITIALISATION.

i) Signal.

Le programme demande le nom du fichier où se trouve enregistré le signal ainsi que les caractéristiques de celui-ci. C'est à dire est-il composé de valeurs réelles ou entières (par exemple signal échantillonné en sortie d'un convertisseur)? fréquence d'échantillonnage?

Les données indépendamment du nombre de bits du convertisseur (exemple pour un 12 bits ie: 11 bits + signe) seront converties en "réel" pour effectuer la transformée. Il est inutile de recadrer le signal car la transformée en ondelette n'analyse pas la fréquence 0 (ie: ajouter dans ce cas 2048 aux données numériques).

ii) Ondelette.

* module :SUPOND

En fonction des valeurs de l'écart type σ , du cut-off de l'enveloppe ϵ , du nombre de points NP(a=1) de l'ondelette et de la valeur de la fréquence d'échantillonnage f_e du signal, on détermine par la subroutine SUPOND, la valeur du support temporel sur lequel existe l'ondelette. L'ondelette est échantillonnée à la même fréquence que le signal.

En sortie du sous-programme nous restituons par le common /support/ les valeurs extrémales du support de l'ondelette (tmin,tmax).

*module :FREQ.

En fonction des caractéristiques de l'ondelette (NP pour a=1, ω_0), de son support (tmin,tmax) et de la fréquence d'échantillonnage du signal transmise par le common /var/, le sous-programme FREQ calcule le nombre de cycles ANC contenu dans l'ondelette a=1 et la fréquence moyenne FREAN analysée par celle-ci.

$$FREAN = ANC.FE/[(NP.a)-1]$$

iii) Grille.

La transformée en ondelettes est calculée sur une grille fine. Connaissant les paramètres de dilatation de départ (dilatation la plus petite), le nombre de voies par octave ainsi que le nombre total

de voies, nous pouvons engendrer une grille dans le plan (b,a) telle que a varie sur [adep,amax], avec :

$$amax = adep.2^{(NVT-1)}/NVO$$

b. CHARGEMENT DU SIGNAL

Après lecture du fichier signal, la partie nécessaire aux calculs est stockée dans le tableau SI.

Le point milieu de l'ondelette pour $b=0$ étant calé sur le premier échantillon du signal, nous devons réserver la longueur d'une demi ondelette associée à la valeur maximum que peut prendre le paramètre a, de part et d'autre de la zone à analyser.

$$\text{d'où } DIM(SI) = 512 + DIM(OND)$$

c. TEST D'ERREURS

i) Aliasing

Le premier test se rapporte au problème d'aliasing. Ceci intervient lorsque le nombre de points NC par cycle de l'ondelette la moins dilatée ne vérifie pas les conditions de Shannon ($ANC \leq 2$).

Ce test est effectuée lors de l'initialisation de l'ondelette. Un message est imprimé lorsque ANC atteint la valeur 3. Il est néanmoins laissé à l'utilisateur la possibilité de continuer la procédure.

ii) Ondelette trop grande

Après initialisation de la grille, on vérifie la compatibilité de la zone mémoire réservée à l'ondelette avec la taille de l'ondelette la plus dilatée.

Si le test est positif, il y a alors arrêt automatique du programme.

Trois choix sont possibles pour effectuer l'analyse:

- soit réduire le nombre de points de l'ondelette pour $a=1$,
- soit réduire le nombre de voies total,
- soit redimensionner les tableaux, en agissant sur le parametre KONDD.

iii) Temps de départ d'analyse trop court

Lorsque le temps de départ de l'analyse est trop petit (dû à ii), un message s'affiche à l'écran. L'utilisateur a alors la possibilité de prolonger le signal par des zéros.

L'analyse mettra en évidence une discontinuité (convergence des lignes de phase constantes, forme caractéristique de la répartition énergétique) en début de fichier.

d. TRANSFORMEE

* module : GENOND

Il permet d'engendrer l'ondelette pour un paramètre de dilatation donné.

De façon à pouvoir se caler sur les différentes voies, l'échantillonnage de l'ondelette est toujours réalisé à partir du point $t=0$ (de part et d'autre).

Il peut être intéressant dans certains cas de normaliser l'ondelette par sa norme L2, limitant ainsi la dynamique des ondelettes. Il suffit pour cela d'activer la séquence prévue à cet effet dans le module. La transformée devra dans ce cas être non plus normalisée par $a^{-1/2}$ (COEFF) mais par $a^{+1/2}$ (dans le module TRANSFON).

Les parties réelles et imaginaires de l'ondelette sont restituées par les COMMON /TABR/ et /TABI/.

* module : TRANSFOND

Ce module réalise le produit de convolution entre le signal et l'ondelette pour une dilatation fixée (une voie). La normalisation des voies est effectuée dans ce module (cf remarque sur COEFF ci dessus).

Les parties réelles et imaginaires de la transformée correspondant au paramètre de dilatation courant, sont restituées par les COMMON /FONCTR/ et /FONCTI/ sur 512 points.

* module : MODPHAS

Il permet de convertir les parties réelles et imaginaires de la transformée en module et phase. Il est préférable d'introduire un cut-off sur le module pour la phase. En effet, celle-ci n'est pas définie pour des valeurs du module trop petites.

Les modules et phases sont restitués par les COMMON /FONCTR/ et /FONCTI/.

e. STOCKAGE DES RESULTATS DE L'ANALYSE

Le choix est donné à l'utilisateur de stocker le module, la phase ou les deux.

La longueur de l'enregistrement est de 512. Le remplissage s'effectue par colonnes. Le premier enregistrement contient les données de l'analyse qui pourront être nécessaires à la représentation graphique: période d'échantillonnage, nombre de points par voie, temps de départ de l'analyse, nombre total de voies, paramètre de dilatation de départ, nombre de voies par octave, fréquence équivalente à $a=1$.

Les paramètres de l'analyse, ainsi que les correspondances entre voies et fréquence moyenne en Hz, sont stockés dans le fichier PARAM.DAT.

3. Génération de signaux tests

Le programme auxiliaire SIGNAL permet d'engendrer des signaux de tests. Un menu présente les 6 choix de signaux possibles:

- impulsion de Dirac
- signal composé de deux signaux carrés espacés de 100.Tech
- sinusoïde
- transition entre deux sinusoïdes de 100Hz et 200Hz.
- somme de deux sinusoïdes de fréquences 100 et 200Hz
- transition entre un Dirac et une sinusoïde de fréquence 150Hz

La fréquence d'échantillonnage est modifiable. Le programme restitue le temps de début d'analyse à utiliser dans ONDEL.

4. Exemple d'analyse sur les fonctions tests

Les figures qui suivent présentent les modules et phases des transformées en ondelettes des signaux tests. Chaque figure est accompagnée du listing du fichier PARAM contenant les paramètres de l'analyse.

La sinusoïde est représentée à deux instants d'analyse différents, le second débutant au temps $t=0$. On met en évidence la discontinuité due au prolongement par zéro du signal.

- fig 1.a: module de la transformée d'un Dirac
- fig 1.b: phase de la transformée d'un Dirac
- fig 2.a: module de la transformée de 2 signaux carrés
- fig 2.b: phase de la transformée de 2 signaux carrés
- fig 3.a: module de la transformée d'une sinusoïde
- fig 3.b: phase de la transformée d'une sinusoïde
- fig 4.a: module de la transformée d'une sinusoïde à $t=0$
- fig 4.b: phase de la transformée d'une sinusoïde à $t=0$
- fig 5.a: module de la transformée d'une transition entre 2 sinusoïdes
- fig 5.b: phase de la transformée d'une transition entre 2 sinusoïdes
- fig 6.a: module de la transformée de la somme de 2 sinusoïdes
- fig 6.b: phase de la transformée de la somme de 2 sinusoïdes
- fig 7.a: module de la transformée de la transition entre un Dirac et une sinusoïde
- fig 7.b: phase de la transformée de la transition entre un Dirac et une sinusoïde

```

NCM DU FICHIER SIGNAL:
SIGNAL.DAT
SIGNAL REEL (1) OU ENTIER (0) :
  1
----- PARAMETRES DE L'ONDELETTE
VALEUR DE L'ECART TYPE DE LA GAUSSIENNE ?
  1.000000
PULSATION PROPRE DE L'ONDELETTE ?
  5.000000
VALEUR MINIMUM POUR LAQUELLE L'ENVELOPPE EST NULLE ?
  9.9999998E-03
NOMBRE DE POINTS DE L'ONDELETTE POUR A=1 ?
  101
FREQUENCE D'ECHANTILLONNAGE DU SIGNAL (EN HZ) ?
  3000.000
---- FREQUENCE MOY. D'ANALYSE POUR A=1:   144.9180   HZ
PARAMETRE DE DILATATION DE DEPART ?
  0.2500000
NOMBRE DE VOIES PAR OCTAVE ?
  8
NOMBRE TOTAL DE VOIES ?
  40
TEMPS DE DEPART DE L'ANALYSE (EN SECONDES) ?
  0.4100000
SAUVEGARDE FICHIER DU MODULE (1) OU DE LA PHASE (2) OU DES DEUX (3) ?
  1
----- VOIE NUMERO:           1
DILATATION:  0.2500000   FREQ.MOYENNE:   603.8250
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           2
DILATATION:  0.2726269   FREQ.MOYENNE:   536.7333
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           3
DILATATION:  0.2973018   FREQ.MOYENNE:   499.7172
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           4
DILATATION:  0.3242099   FREQ.MOYENNE:   452.8687
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           5
DILATATION:  0.3535534   FREQ.MOYENNE:   414.0514
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           6
DILATATION:  0.3855527   FREQ.MOYENNE:   381.3632
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           7
DILATATION:  0.4204482   FREQ.MOYENNE:   353.4585
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           8
DILATATION:  0.4585020   FREQ.MOYENNE:   322.0400
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:           9
DILATATION:  0.5000000   FREQ.MOYENNE:   289.8360
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:          10
DILATATION:  0.5452539   FREQ.MOYENNE:   268.3667
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:          11
DILATATION:  0.5946035   FREQ.MOYENNE:   245.6237
MOD= ,PHAS=  0.0000000E+00  0.0000000E+00
----- VOIE NUMERO:          12

```

fig: 1.a. Module de la transformée d'un Dirac.




fig 1.b. Phase de la transformée d'un Dirac




fig: 2.a. Module de la transformée de 2 signaux carrés




fig: 2.b . Phase de la transformée de 2 signaux carrés .




Fig: 5.a . Module de la transformée de la transition entre 2 sinusoides .




fig: 5. b : Phase de la transition entre 2 sinusoides





fig: 6. a . Module de la transformée de la somme de 2 sinusoides .

$\times 1.E+03$



\log_2

fig: 6.6 : Phase de la transformée : la somme de 2 sinusoides

$\times 1.E+03$




fig 7. a. Module de la transformée de la transition entre un Dirac et une sinusoïde.







fig: 7.6. Phase de la transformée de la transition entre un Dirac et une sinusoïde.



ANNEXE



Partie imaginaire de l'ondelette de Morlet.



Partie réelle de l'ondelette de Morlet.

Parties réelles et imaginaire de l'ondelette Morlet pour $NP=101$, $\epsilon=10^{-2}$, $a=10$, $\omega_0=5$

5. Programmes sources des logiciels:

- ONDEL.FOR

- SIGNAL.FOR

```

C-----
C SARACCO G. 16-6-87          ** ONDEL.FOR **
C
C PROG: ONDEL.FOR ; CALCUL DE LA TRANSFORMEE EN ONDELETTE D'UN SIGNAL
C          ARBITRAIRE CONTENU DANS LE FICHIER REEL 'SIGNAL.DAT'
C LA TRANSFORMEE EST EFFECTUEE SUR 512 POINTS ET STOCKEE DANS LE FICHIER
C 'TEMFRE.DAT' (MATRICE IMAGE 512 X NBR DE VOIES)
C L'ONDELETTE UTILISEE ICI EST  $s(t) = \exp(-(t**2)/2*SIG)\exp(iW0t)$ 
C (ONDELETTE TYPE MORLET)
C-----
C
C- SUBROUTINE FREQ
C- RELATION ENTRE DILATATION ET FREQUENCE D'ANALYSE
C   ARGUMENTS: A  PARAMETRE DE DILATATION
C               FE FREQUENCE D'ECHANTILLONAGE DU SIGNAL (PAR COMMON)
C               NP NOMBRE DE POINTS DE L'ONDELETTE POUR A=1
C               WO PULSATION PROPRE DE L'ONDELETTE
C               TMIN ET TMAX (SUPPORT POUR A=1) PAR COMMON
C   RESTITUE FREAN: FREQUENCE MOYENNE D'ANALYSE
C               ANC  : NBR DE CYCLES DE L'ONDELETTE
C
C       SUBROUTINE FREQ(A, NP, WO, FREAN, ANC)
C       COMMON/SUPPORT/TMIN, TMAX
C       COMMON/VAR/FE
C       PI2=ATAN(1.)*8.
C----- NOMBRE DE CYCLES DANS L'ONDELETTE ---
C       ANC=(TMAX-TMIN)*WO/PI2
C----- NOMBRE DE POINTS DE L'ONDELETTE -----
C       NBRPNT=NINT(NP*A)
C----- FREQUENCES ANALYSEES -----
C       FREAN=ANC*FE/(NBRPNT-1)
C       RETURN
C       END
C
C- FUNCTION
C- GENERATION DE LA GAUSSIENNE (ENVELOPPE DE L'ONDELETTE)
C   ARGUMENTS: TE  VARIABLE TEMPORELLE
C               SIG ECART TYPE, FIXE LE (DELTA F/F)
C
C       FUNCTION ENV(TE, SIG)
C       ENV=EXP((-TE**2)/2*SIG)
C       RETURN
C       END
C
C - SUBROUTINE SUPOND
C - SUPPORT TEMPOREL DE L'ONDELETTE POUR UNE DILATATION DE 1
C   ARGUMENTS: SIG  ECART TYPE ASSOCIE A L'ONDELETTE
C               BMIN CUTOFF NUMERIQUE
C               FE   FREQUENCE D'ECHANTILLONAGE (FIXE LE PAS)
C   RESTITUE PAR LE COMMON LES BORNES TMIN ET TMAX
C
C       SUBROUTINE SUPOND(SIG, BMIN, FE)
C       COMMON/SUPPORT/TMIN, TMAX
C       PAS=1./FE
C       TE=0.
1      ONDA=ENV(TE, SIG)
C       IF(ONDA.GT.BMIN) THEN
C           TE=TE+PAS
C           GOTO 1
C       ELSE

```

```

      TMAX=TE
    END IF
    TE=0.
2    ONDA=ENV(TE,SIG)
    IF(ONDA.GT.BMIN)THEN
      TE=TE-PAS
      GOTO 2
    ELSE
      TMIN=TE
    END IF
    RETURN
  END

C
C - SUBROUTINE GENOND
C - CONSTRUCTION DE L'ONDELETTE TYPE MORLET
C - PARTIE RELLE (RE)- PARTIE IMAGINAIRE (IM)
C - SUPPORT(A)=(TMAX-TMIN)*A, NP(A)=NINT(NP(A=1)*A)
C   ARGUMENTS: WO PULSATION DE L'ONDELETTE POUR A=1
C               SIG ECART TYPE DE LA GAUSSIENNE
C               A   VALEUR DU PARAMETRE DE DILATATION
C               NP NOMBRE DE POINTS DE L'ONDELETTE POUR A=1
C               NBRPNT NRE DE PTS DE L'ONDEL. POUR UN PARAMETRE A DONNE
C               KOND LONGUEUR MAX DE L'ONDELETTE
C RESTITUE PAR LE COMMON/TABR/ LES TABLEAUX REWAV ET/TABI/ IMWAV
C CONTENANT RESPECTIVEMENT LES PARTIES REELLES ET IMAGINAIRES
C DE L'ONDELETTE
C
      SUBROUTINE GENOND(WO,SIG,A,NP,NBRPNT,KOND)
      COMMON/TABR/REWAV/TABI/IMWAV
      COMMON /SUPPORT/TMIN,TMAX
      COMMON/VAR/FE
      REAL IMWAV(1),REWAV(1)

C
C----- NOMBRE DE POINTS DE L'ONDELETTE -----
      NBRPNT=NINT(NP*A)
C--- PAS D'ECHANTILLONNAGE DE L'ONDELETTE ---
      PAS=(TMAX-TMIN)/FLOAT(NBRPNT-1)
C-----
      DO 2 I=1,KOND
        REWAV(I)=0.
        IMWAV(I)=0.
2      CONTINUE
C----- POINT MILIEU-----
C 3000 IE: KOND POINTS SONT RESERVES POUR LE STOCKAGE DE L'ONDELETTE
C LE POINT MILIEU SERA TOUJOURS EN (KOND/2) POUR CALAGE DES VOIES
      KM=KOND/2
      TO=0.
      ONDA=ENV(TO,SIG)
      REWAV(KM)=ONDA
      IMWAV(KM)=0.
C-----
      T=0.
      ANORL2=0.   ! normalisation L2
      DO 1 K=1,NINT((NBRPNT-1)/2.)
        TO=T+PAS*FLOAT(K)
        ONDA=ENV(TO,SIG)
        ANORL2=ANORL2+ONDA*ONDA
        REWAV(KM+K)=ONDA*COS(WO*TO)
        REWAV(KM-K)=REWAV(KM+K)
        IMWAV(KM+K)=ONDA*SIN(WO*TO)

```

```

      TMAX=TE
    END IF
    TE=0.
2    ONDA=ENV(TE,SIG)
    IF(ONDA.GT.BMIN)THEN
      TE=TE-PAS
      GOTO 2
    ELSE
      TMIN=TE
    END IF
    RETURN
  END

C
C - SUBROUTINE GENOND
C - CONSTRUCTION DE L'ONDELETTE TYPE MORLET
C - PARTIE RELLE (RE)- PARTIE IMAGINAIRE (IM)
C - SUPPORT(A)=(TMAX-TMIN)*A, NP(A)=NINT(NP(A=1)*A)
C   ARGUMENTS: WO PULSATION DE L'ONDELETTE POUR A=1
C               SIG ECART TYPE DE LA GAUSSIENNE
C               A   VALEUR DU PARAMETRE DE DILATATION
C               NP NOMBRE DE POINTS DE L'ONDELETTE POUR A=1
C               NBRPNT NBRE DE PTS DE L'ONDEL. POUR UN PARAMETRE A DONNE
C               KOND LONGUEUR MAX DE L'ONDELETTE
C RESTITUE PAR LE COMMON/TABR/ LES TABLEUX REWAV ET/TABI/ IMWAV
C CONTENANT RESPECTIVEMENT LES PARTIES REELLES ET IMAGINAIRES
C DE L'ONDELETTE
C
      SUBROUTINE GENOND(WO,SIG,A,NP,NBRPNT,KOND)
      COMMON/TABR/REWAV/TABI/IMWAV
      COMMON /SUPPORT/TMIN,TMAX
      COMMON/VAR/FE
      REAL IMWAV(1),REWAV(1)

C
C----- NOMBRE DE POINTS DE L'ONDELETTE -----
      NBRPNT=NINT(NP*A)
C--- PAS D'ECHANTILLONNAGE DE L'ONDELETTE --
      PAS=(TMAX-TMIN)/FLOAT(NBRPNT-1)
C-----
      DO 2 I=1,KOND
        REWAV(I)=0.
        IMWAV(I)=0.
2      CONTINUE
C----- POINT MILIEU-----
C 3000 IE: KOND POINTS SONT RESERVES POUR LE STOCKAGE DE L'ONDELETTE
C LE POINT MILIEU SERA TOUJOURS EN (KOND/2) POUR CALAGE DES VOIES
      KM=KOND/2
      TO=0.
      ONDA=ENV(TO,SIG)
      REWAV(KM)=ONDA
      IMWAV(KM)=0.
C-----
      T=0.
      ANORL2=0.   ! normalisation L2
      DO 1 K=1,NINT((NBRPNT-1)/2.)
        TO=T+PAS*FLOAT(K)
        ONDA=ENV(TO,SIG)
        ANORL2=ANORL2+ONDA*ONDA
        REWAV(KM+K)=ONDA*COS(WO*TO)
        REWAV(KM-K)=REWAV(KM+K)
        IMWAV(KM+K)=ONDA*SIN(WO*TO)

```

```

          IMWAV(KM-K)=-IMWAV(KM+K)
1      CONTINUE
C----- NORMALISATION-----
C      DO 3 I=1,KOND
C          IMWAV(I)=IMWAV(I)/ANORL2
C          REWAV(I)=REWAV(I)/ANORL2
C3     CONTINUE
        RETURN
        END

C
C-----
C SUBROUTINE TRANSFOND : TRANSFORMATION EN ONDELETTES.
C ARGUMENTS: A      PARAMETRE DE DILATATION COURANT
C            NBRPNT NOMBRE DE POINT DE L'ONDELETTE POUR A
C            IDEB   INDEX DU 1ER POINT D'ANALYSE DU SIGNAL
C            IFIN   "      DERNIER      "      "
C            KOND   LONGUEUR MAX DE L'ONDELETTE
C            REWAV  ET IMWAV CONTIENNENT L'ONDELETTE PAR LE COMMON
C            SI     SIGNAL PAR COMMON/FSI/
C RESTITUE TFRE ET TFIM CONTENANT LES PARTIES REELLES ET IMAGINAIRES
C DE LA TRANSFORMEE PAR LES COMMON/FONCTR/ ET /FONCTI/.
C
      SUBROUTINE TRANSFOND(A,NBRPNT,IDEB,IFIN,KOND)
      REAL IMWAV(1),TFRE(1),TFIM(1),SI(1),REWAV(1)
      COMMON /FSI/SI/FONCTR/TFRE/FONCTI/TFIM
      COMMON/TABR/REWAV/TABI/IMWAV
      KM=KOND/2 !POINT MILIEU
C-----
      COEFF=1./SQRT(A)
      INDICE=NINT((NBRPNT-1)/2.)
      DO 1 I=IDEB,IFIN
          TRRE=0. !INITIALISATION DU PRODUIT DE CONVOLUTION
          TRIM=0.
          DO 2 L=I-INDICE,I+INDICE
              TRRE=TRRE+SI(L)*REWAV(L-I+KM+1)
              TRIM=TRIM+SI(L)*IMWAV(L-I+KM+1)
2      CONTINUE
          TFRE(I-IDEB+1)=TRRE*COEFF
          TFIM(I-IDEB+1)=TRIM*COEFF
1      CONTINUE
      RETURN
      END

C
C-----
C SUBROUTINE MODPHAS
C ARGUMENTS:TFRE,TFIM TABLEAUX CONTENANT RESPECTIVEMENT LES PARTIES
C REELLES ET IMAGINAIRES DE LA TRANSFORMEE (TRANSMIS PAR COMMON).
C CUTOFF VALEUR LIMITE DU MODULE
C KNP DIMENSION TEMPORELLE DE LA TRANSFORMEE
C TFRE ET TFIM SONT CONVERTIS EN MODULE ET PHASE(MOD 2*PI), RESTITUES
C PAR LES COMMON /FONCTR/ ET /FONCTI/.
C
      SUBROUTINE MODPHAS(CUTOFF,KNP)
      DIMENSION TFRE(1),TFIM(1)
      COMMON/FONCTR/TFRE/FONCTI/TFIM
      DO I=1,KNP
          CR=TFRE(I)
          CI=TFIM(I)
          TFRE(I)=SQRT(CR*CR+CI*CI)
          IF(TFRE(I).GT.CUTOFF)THEN

```

```

        TFIM(I)=ATAN2(-CI,CR)
    ELSE
        TFIM(I)=0.
    END IF
END DO
RETURN
END

C
C-----
C   PROGRAMME PRINCIPAL
C-----
        PARAMETER (KNP=512, KOND=3000, KRCL=256, KSI=4512)
C***** KSI= LONGUEUR DU SIGNAL ANALYSE : KSI >=KNP+KOND *****
C   KOND=LONGUEUR DE L"ONDELETTE
C   KRCL=LONGUEUR DE L"ENREGISTREMENT DU SIGNAL
C***** KNP= LONGUEUR DE L"ENREGISTREMENT (TEMPS) DE LA TRANSFORMEE
C
        REAL IMWAV(KOND)
        DIMENSION TSMOD(KOND),TSIM(KOND),TSRE(KOND),SI(KSI)
        DATA KKRCL,KKNP/KRCL,KNP/
        CHARACTER NOMSIG*13, NOMMOD*13, NOMPHA*13
        DIMENSION TAB(KNP,KNP),TAB1(KNP,KNP),REWAV(KOND)
        DIMENSION ISIGNA(KRCL),SIGNAL(KRCL),TFRE(KNP),TFIM(KNP)
        COMMON /FSI/SI/FONCTR/TFRE/FONCTI/TFIM
        COMMON/TABR/REWAV/TABI/IMWAV
        COMMON /SUPPORT/TMIN,TMAX
        COMMON/VAR/FE
C FICHER CONTENANT LES PARAMETRES DE L'ANALYSE
        OPEN(UNIT=4,NAME='PARAM.DAT',TYPE='NEW')
C-----
C----- INITIALISATION DES PARAMETRES DE L'ONDELETTE ET DE LA GRILLE
C-----
        TYPE 79
        WRITE(4,79)
79      FORMAT(1H$, 'NOM DU FICHER SIGNAL: ')
        ACCEPT 78,NOMSIG
        WRITE(4,78)NOMSIG
78      FORMAT(A)
        TYPE 77
        WRITE(4,77)
77      FORMAT(1H$, 'SIGNAL REEL (1) OU ENTIER (0) : ')
        ACCEPT *,ISIGN
        WRITE(4,*)ISIGN
        TYPE *, '----- PARAMETRES DE L"ONDELETTE'
        WRITE(4,*) '----- PARAMETRES DE L"ONDELETTE'
        TYPE 80
        WRITE(4,80)
80      FORMAT(1H$, 'VALEUR DE L"ECART TYPE DE LA GAUSSIENNE ?')
        ACCEPT *,SIG
        WRITE(4,*)SIG
        TYPE 81
        WRITE(4,81)
81      FORMAT(1H$, 'PULSATION PROPRE DE L"ONDELETTE ?')
        ACCEPT *,W0
        WRITE(4,*)W0
        TYPE 89
        WRITE(4,89)
89      FORMAT(1H$, 'VALEUR MINIMUM POUR LAQUELLE L"ENVELOPPE EST NULLE
        ACCEPT *,BMIN
        WRITE(4,*)BMIN

```



```

TYPE 82
WRITE(4,82)
82  FORMAT(1H$, 'NOMBRE DE POINTS DE L"ONDELETTE POUR A=1 ?')
    ACCEPT *, NP
    WRITE(4,*)NP
    TYPE 83
    WRITE(4,83)
83  FORMAT(1H$, 'FREQUENCE D"ECHANTILLONNAGE DU SIGNAL (EN HZ) ?')
    ACCEPT *, FE
    WRITE(4,*)FE
C  SUPPORT TEMPOREL DE L'ONDELETTE POUR A=1
    CALL SUPOND(SIG,BMIN,FE)
C  RELATION ENTRE A=1 ET FREQUENCE D'ANALYSE
    CALL FREQ(1.,NP,WO,FREAN,ANC)
    TYPE *, '----- FREQUENCE MOY. D"ANALYSE POUR A=1: ',FREAN,'HZ'
    WRITE(4,*)'----- FREQUENCE MOY. D"ANALYSE POUR A=1: ',FREAN,'HZ'
    FREA1=FREAN
    TYPE 85
    WRITE(4,85)
85  FORMAT(1H$, 'PARAMETRE DE DILATATION DE DEPART ?')
    ACCEPT *, ADEP
    WRITE(4,*)ADEP
C  NBR DE POINTS PAR CYCLES (POUR SIGNALER L'ALIASING) POUR A=ADEP
    NPTCY=INT(NP*ADEP/ANC)
    IF(NPTCY.LT.3)THEN
        TYPE *, ' '
        TYPE *, '***** ATTENTION DANGER D"ALIASING *****'
        TYPE *, '***** L"ONDELETTE CONTIENT ',NPTCY,' POINTS PAR CYCLE'
        TYPE *, '***** POUR A= ',ADEP
        TYPE *, '          CONTINUER :1, ARRETER: 0'
        WRITE(4,*)' '
        WRITE(4,*)'***** ATTENTION DANGER D"ALIASING *****'
        WRITE(4,*)' L"ONDELETTE CONTIENT ',NPTCY,' POINTS PAR CYCLE'
        WRITE(4,*)'          POUR A= ',ADEP
        WRITE(4,*)'          CONTINUER :1, ARRETER: 0'
        ACCEPT *, ICONT
        WRITE(4,*)ICONT
        TYPE *, ' '
        WRITE(4,*)' '
        IF(ICONT.EQ.0)THEN
            STOP
        END IF
    END IF
    TYPE 84
    WRITE(4,84)
84  FORMAT(1H$, 'NOMBRE DE VOIES PAR OCTAVE ?')
    ACCEPT *, NVO
    WRITE(4,*)NVO
    TYPE 86
    WRITE(4,86)
86  FORMAT(1H$, 'NOMBRE TOTAL DE VOIES ?')
    ACCEPT *, NVT
    WRITE(4,*)NVT
C-----
C  VALEUR DE A MAXIMUM ET LONGUEUR DE L'ONDELETTE LA PLUS DILATEE
    AMAX=ADEP*2**((NVT-1)/FLOAT(NVO))
    LMAX=NP*AMAX
    IF(LMAX/2..GT.FLOAT(KOND))THEN
        TYPE *, ' '
        WRITE(4,*)' '

```

```

        TYPE 100
        WRITE(4,100)
100      FORMAT(1H$, '***** ONDELETTE TROP GRANDE *****')
        STOP
    END IF
C ECHANTILLON DE DEPART DANS LE FICHIER POUR ANALYSE
    TYPE 87
    WRITE(4,87)
87      FORMAT(1H$, 'TEMPS DE DEPART DE L"ANALYSE (EN SECONDES) ?')
        ACCEPT *,TDEP
        WRITE(4,*)TDEP
        IECHDE=NINT(TDEP*FE)
C DEMI ONDELETTE LA PLUS GRANDE
        IDO=NINT((LMAX+1)/2.)
        IF(IECHDE-IDO.LE.0)THEN
            TYPE *, ' '
            WRITE(4,*)' '
            TYPE 101
            WRITE(4,101)
101     FORMAT(X, '***** TEMPS DE DEPART D"ANALYSE TROP PETIT *****')
            TYPE *, '***** LE SIGNAL PEUT ETRE PROLONGE PAR DES ZEROS'
            WRITE(4,*)'***** LE SIGNAL PEUT ETRE PROLONGE PAR DES ZEROS'
            TYPE 102
            WRITE(4,102)
102     FORMAT(1H$, '          CONTINUER :1, ARRETER :0 ')
            ACCEPT *,ICOD
            WRITE(4,*)ICOD
            IF(ICOD.EQ.0)THEN
                STOP
            END IF
        ELSE
            ICOD=2
        END IF
        TYPE 90
        WRITE(4,90)
90      FORMAT(1H$, 'SAUVEGARDE FICHIER DU MODULE (1) OU DE LA PHASE (2)
1      ' OU DES DEUX (3) ?')
        ACCEPT*,ITYPE
        WRITE(4,*)ITYPE
        IF(ITYPE.EQ.1.OR.ITYPE.EQ.3)THEN      ! OUVERTURE FICHIER MODULE
            TYPE 103
            WRITE(4,103)
103     FORMAT(1H$, 'NOM DU FICHIER MODULE ?')
            ACCEPT 104,NOMMOD
            WRITE(4,104)NOMMOD
104     FORMAT(A)
            OPEN(UNIT=3,ACCESS='DIRECT',RECL=KKNP,NAME=NOMMOD,STATUS='NEW')
        END IF
        IF(ITYPE.GE.2)THEN      ! OUVERTURE FICHIER PHASE
            TYPE 105
            WRITE(4,105)
105     FORMAT(1H$, 'NOM DU FICHIER PHASE ?')
            ACCEPT 104,NOMPHA
            WRITE(4,104)NOMPHA
            OPEN(UNIT=2,ACCESS='DIRECT',RECL=KKNP,NAME=NOMPHA,STATUS='NEW')
            TYPE 88
            WRITE(4,88)
88      FORMAT(1H$, 'VALEUR DU CUTOFF POUR LA REPRESENTATION GRAPH. ?')
            ACCEPT*,CUTOFF
            WRITE(4,*)CUTOFF

```

```

      END IF
C BORNES DE REMPLISSAGE DU TABLEAU SI ET INITIALISATION DE SI A ZERO
      DO 35 I=1,KSI
35      SI(I)=0.
          KBINF=(KSI-KNP)/2
          KBSUP=KBINF+KNP
          INMIN=KBINF-IDO
          INMAX=KBSUP+IDO
C LECTURE DU SIGNAL DANS LE FICHIER 'SIGNAL' ET REMPLISSAGE DE SI
      OPEN(UNIT=1,STATUS='OLD',ACCESS='DIRECT',RECL=KKRCL,NAME=NOMSIG
      IF(ICOD.EQ.1)THEN !LECTURE DEPUIS LE DEBUT ET PROLDNG. PAR 0
          NUMENR=1 !NUMERO DE L'ENREGISTREMENT DE DEPART
          IPRPNT=0
          NPTMAN=IDO-IECHDE !NBR DE POINTS A PROLONGER PAR 0
          INMIN=INMIN+NPTMAN
      ELSE
          NUMENR=(IECHDE-IDO)/KRCL !NUMERO DE L'ENREGISTREMENT DE DEPART
          IF(NUMENR*KRCL.LT.(IECHDE-IDO))NUMENR=NUMENR+1
          IPRPNT=(IECHDE-IDO)-KRCL*(NUMENR-1) !NUMERO DU 1ER PNT DANS L'E.
          NPTMAN=0
      END IF
      IF(ISIGN.EQ.1)THEN
          READ(1'NUMENR,ERR=1000)SIGNAL
          DO 50 I=IPRPNT,KRCL
              SI(INMIN-IPRPNT+I)=SIGNAL(I)
50          CONTINUE
      ELSE
          READ(1'NUMENR,ERR=1000)ISIGNA
          DO 51 I=IPRPNT,KRCL
              SI(INMIN-IPRPNT+I)=FLOAT(ISIGNA(I))
51          CONTINUE
      END IF
      NPL=KNP+2*(IDO)-NPTMAN ! NBR DE POINTS A LIRE DANS LE FICHIER
      NLU=KRCL-IPRPNT !NOMBRE DE POINTS DEJA REMPLIS
      NRL=NPL-NLU !NOMBRE DE POINTS RESTANT A LIRE
      NERL=NRL/KRCL !NOMBRE D'ENREG RESTANT A LIRE
      IF(NERL*KRCL.LT.NERL)NERL=NERL+1
      IF(ISIGN.EQ.1)THEN
          DO 60 I=1,NERL-1
              READ(1'NUMENR+I,ERR=1000)SIGNAL
              DO 60 J=1,KRCL
                  SI(INMIN+NLU+J+KRCL*(I-1))=SIGNAL(J)
60          CONTINUE
      ELSE
          DO 61 I=1,NERL-1
              READ(1'NUMENR+I,ERR=1000)ISIGNA
              DO 61 J=1,KRCL
                  SI(INMIN+NLU+J+KRCL*(I-1))=FLOAT(ISIGNA(J))
61          CONTINUE
      END IF
      NLU1=NLU+KRCL*(NERL-1)
      NRL=NPL-NLU1
      IF(ISIGN.EQ.1)THEN
          READ(1'NUMENR+NERL,ERR=1000)SIGNAL
          DO 70 I=1,NRL
              SI(INMIN+NLU+KRCL*(NERL-1)+I)=SIGNAL(I)
70          CONTINUE
      ELSE
          READ(1'NUMENR+NERL,ERR=1000)ISIGNA
          DO 71 I=1,NRL
              SI(INMIN+NLU+KRCL*(NERL-1)+I)=FLOAT(ISIGNA(I))
71          CONTINUE
      END IF

```

```

      END IF
C
C ECRITURE DE LA TRANSFORMEE DU SIGNAL SOUS FORME DE CARTE (KNP X NVT)
C -----
C TRANSFORMEE VOIES PAR VOIES
C-----
      DO 20 I=1,NVT
        A=ADEP*2**((I-1)/FLOAT(NVO))
C RELATION ENTRE A ET FREQUENCE D'ANALYSE
        CALL FREQ(A,NP,WO,FREAN,ANC)
        WRITE(4,*)'----- VOIE NUMERO: ',I
        WRITE(4,*)'DILATATION: ',A,' FREQ.MOYENNE: ',FREAN
C GENERATION DE L'ONDELETTE
        CALL GENOND(WO,SIG,A,NP,NBRPNT,KOND)
        IDEB=KBINF
        IFIN=KBSUP
C TRANSFORMEE DU SIGNAL
        CALL TRANSFOND(A,NBRPNT,IDEB,IFIN,KOND)
C CONVERSION PARTIES REELLE ET IMAGINAIRE EN MODULE ET PHASE
C CODE SUR 512X512 ,IE: KNP X KNP
        CALL MODPHAS(CUTOFF,KNP)
        DO 30 IT=1,KNP
          IF(ITYPE.EQ.1.OR.ITYPE.EQ.3)THEN !SAUVEGARDE DU MODULE
            TAB(I,IT)=TFRE(IT)
          ENDIF
          IF(ITYPE.GE.2)THEN
            TAB1(I,IT)=TFIM(IT) !SAUVEGARDE DE LA PHASE
          END IF
30      CONTINUE
20      CONTINUE
        DT=1/FE
        TMAX=FLOAT(KNP)*DT !DUREE DE LA FENETRE D'ANALYSE
C ENTETE DU FICHER RESULTAT
C PERIOD,ECH,NBR POINTS/VOIE,TEMPS DEPART,NBR VOIES,DILAT DEP.,
C NBR VOIES/OCT,FREQ MOYENNE POUR A=1
        IF(ITYPE.EQ.1.OR.ITYPE.EQ.3)THEN
          WRITE(3'1)DT,KNP,TDEP,NVT,ADEP,NVO,FREA1
C IMAGE CALCULEE
          DO L=2,KNP+1
            WRITE(3'L)(TAB(M,L-1),M=1,NVT) !REPLISSAGE EN COLONNES
          END DO
        END IF
        IF(ITYPE.GE.2)THEN
          WRITE(2'1)DT,KNP,TDEP,NVT,ADEP,NVO,FREA1
C IMAGE CALCULEE
          DO L=2,KNP+1
            WRITE(2'L)(TAB1(M,L-1),M=1,NVT) !REPLISSAGE EN COLONNES
          END DO
        END IF
        GO TO 1001
1000  TYPE *, '*** ERREUR DE LECTURE FICHER SIGNAL ***'
1001  CLOSE(UNIT=1)
        CLOSE(UNIT=2)
        CLOSE(UNIT=3)
        CALL EXIT
      END

```

SIGNAL.FOR

C GENERATION D'UN FICHER CONTENANT UN SIGNAL ECHANTILLONNE
 C LES ECHANTILLONS SONT CODES EN REELS
 C L'ACCES EST DIRECT PAR ENREGISTREMENTS DE 256 ECHANTILLONS
 C

 SUBROUTINE DIRAC

C GENERATION D'UN FICHER NE CONTENANT QU'UN DIRAC EN 1512

C TDEP=1250/FECH s
 COMMON SIGNAL(256)
 DO I=1,256
 SIGNAL(I)=0.
 END DO
 DO J=1,5
 WRITE(1'J)SIGNAL
 END DO
 SIGNAL(232)=100.
 WRITE(1'6)SIGNAL
 SIGNAL(232)=0.
 DO K=7,12
 WRITE(1'K)SIGNAL
 END DO
 RETURN
 END

C-----
 C SUBROUTINE DBLCAR
 C FONCTION COMPOSEE DE 2 SIGNAUX CARRES ESPACES DE 100*TECHs
 C LE PREMIER AU TPS 1330/FECH s
 C TDEP=1250/FECH s
 COMMON SIGNAL(256)
 DO 1 I=1,256
 SIGNAL(I)=0.
 1 CONTINUE
 DO 2 J=1,5
 WRITE(1'J)SIGNAL
 2 CONTINUE
 DO I=50,100
 SIGNAL(I)=100.
 END DO
 DO I=200,256
 SIGNAL(I)=100.
 END DO
 C
 WRITE(1'6)SIGNAL
 DO I=50,256
 SIGNAL(I)=0.
 END DO
 DO 3 J=7,12
 WRITE(1'J)SIGNAL
 3 CONTINUE
 RETURN
 END

C-----
 C SUBROUTINE SINUS
 C SINUSOIDE DE FREQUENCE FE HZ, ECHANTILLONNEE A FECH HZ
 C
 COMMON SIGNAL(256)/FREQ/FECH
 TECH=1/FECH
 WRITE(5,101)
 101 FORMAT(1H#, 'FREQUENCE EN Hz DE LA SINUSOIDE ?')
 READ(5,*)FE
 PIZ=ATAN(1.)*8.

```

DO 1 J=1,16
DO 2 I=1,256
  K=(I-1)+256*(J-1)
  SIGNAL(I)=SIN(PI2*FE*K*TECH)
2 CONTINUE
1 WRITE(1'J)SIGNAL
  RETURN
  END

```

```

-----
C SUBROUTINE TRANSI
C
C TRANSITION ENTRE 2 SINUSOIDES DE FREQUENCE F1 HZ ET F2 HZ
C (PAR EXEMPLE 100 Hz ET 200 Hz)
C ECHANTILLONNEES A FECH HZ
C LA TRANSITION EST AU DEBUT DU 8eme BLOC, SOIT AU TPS 1792/FECH s
C TDEP=1536/FECH s
C
COMMON SIGNAL(256)/FREQ/FECH
TECH=1/FECH
WRITE(5,104)
104 FORMAT(1H$, 'FREQUENCE DE LA 1ere SINUSOIDE ?')
  READ(5,*)F1
  WRITE(5,105)
105 FORMAT(1H$, 'FREQUENCE DE LA 2eme SINUSOIDE ?')
  READ(5,*)F2
  PI2=ATAN(1.)*8.
  DO 1 J=1,16
  IF(J.LT.8) THEN
    F=F1
  ELSE
    F=F2
  END IF
  DO 2 I=1,256
    K=(I-1)+256*(J-1)
    SIGNAL(I)=SIN(PI2*F*K*TECH)
2 CONTINUE
1 WRITE(1'J)SIGNAL
  RETURN
  END

```

```

-----
C SUBROUTINE SOMSIN
C SOMME DE 2 SINUSOIDES DE FREQUENCES DIFFERENTES
C A F1 HZ ET F2 HZ
C
COMMON SIGNAL(256)/FREQ/FECH
TECH=1/FECH
WRITE(5,102)
102 FORMAT(1H$, 'FREQUENCE DE LA 1ere SINUSOIDE EN Hz ?')
  READ(5,*)F1
  WRITE(5,103)
103 FORMAT(1H$, 'FREQUENCE DE LA 2eme SINUSOIDE EN Hz ?')
  READ(5,*)F2
  PI2=ATAN(1.)*8.
  DO 1 J=1,16
  DO 2 I=1,256
    K=(I-1)+256*(J-1)
    SIGNAL(I)=SIN(PI2*F1*K*TECH)+SIN(PI2*F2*K*TECH)
2 CONTINUE
1 WRITE(1'J)SIGNAL
  RETURN

```


END

C-----
SUBROUTINE DIRSIN
C GENERATION D'UN FICHER CONTENANT UN DIRAC EN 1280 ET UNE SINUSOIDE
C A LA FREQUENCE 150 HZ, ESPACES DE 1*TECH s
C TDEP= 1030/FECH s
COMMON SIGNAL(256)/FREQ/FECH
TECH=1/FECH
PI2=ATAN(1.)*8.
DO I=1,256
SIGNAL(I)=0.
END DO
DO J=1,5
WRITE(1'J)SIGNAL
END DO
SIGNAL(256)=100.
WRITE(1'5)SIGNAL
DO 1 J=1,16
DO 2 I=1,256
K=(I-1)+256*(J-1)
SIGNAL(I)=SIN(PI2*150.*K*TECH)
2 CONTINUE
1 WRITE(1'J+5)SIGNAL
RETURN
END


C-----
C CHOIX DU SIGNAL
COMMON SIGNAL(256)/FREQ/FECH
OPEN(UNIT=1,STATUS='NEW',ACCESS='DIRECT',RECL=256,NAME='SIGNAL.DA
TYPE 81
81 FORMAT(X,'FONCTIONS "TEST"')
TYPE 83
83 FORMAT(1H\$, 'FREQUENCE D"ECHANTILLONNAGE FECH EN HERTZ ?')
ACCEPT *,FECH
WRITE(5,82)
82 FORMAT(X,'QUEL TYPE DE SIGNAUX VOULEZ-VOUS GENERER?:')
WRITE(5,*)'1- UN DIRAC ?'
WRITE(5,*)'2- DEUX SIGNAUX CARRES ESPACEES DE 100/FECH s ?'
WRITE(5,*)'3- UNE SINUSOIDE ?'
WRITE(5,*)'4- TRANSITION ENTRE 2 SINUSOIDES ?'
WRITE(5,*)'5- LA SOMME DE DEUX SINUSOIDES ?'
WRITE(5,*)'6- UN DIRAC ET UNE SINUSOIDE DE FREQ. 150 Hz ?'
WRITE(5,*)'7-FIN '
READ(5,*)ICHOIX
IF((ICHOIX.LT.1).OR.(ICHOIX.GT.7))GOTO 100
GO TO (10,20,30,40,50,60,100),ICHOIX
10 WRITE(5,*)'ANALYSE AU TEMPS: ',1256./FECH,'s'
CALL DIRAC
GOTO 100
20 WRITE(5,*)'ANALYSE AU TEMPS: ',1230./FECH,'s'
CALL DBLCAR
GOTO 100
30 WRITE(5,*)'ANALYSE AU TEMPS: ',1250./FECH,'s'
CALL SINUS
GOTO 100
40 WRITE(5,*)'ANALYSE AU TEMPS: ',1536./FECH,'s'
CALL TRANSI
GOTO 100
50 WRITE(5,*)'ANALYSE AU TEMPS: ',1250./FECH,'s'
CALL SOMSIN

```
GOTO 100
60  WRITE(5,*) 'ANALYSE AU TEMPS: ',1030./FECH,'s'
    CALL DIRSIN
    GOTO 100
100  CLOSE(UNIT=1)
    CALL EXIT
    END
```


ANNEXE



Partie imaginaire de l'ondelette de Morlet.



Partie réelle de l'ondelette de Morlet.

Parties réelles et imaginaires de l'ondelette Morlet pour $NP=101$, $\epsilon=10^{-2}$, $a=10$, $\omega_0=5$

PUBLICATIONS

AGUIRRE J., ESCOBEDO M., PERAL J.C., TCHAMITCHIAN Ph.
Bases d'ondelettes et décompositions atomiques des espaces $H^1(\mathbb{R}^n)$ et $H^1(\mathbb{R}^n \times \mathbb{R}^n)$.
Soumis aux Transactions AMS.

BALSLEV E., GROSSMANN A., PAUL T.
A characterization of dilation analytic operators.
Annales I.H.P. 45, 277 (1986).

DAUBECHIES I., GROSSMANN A., MEYER Y.
Painless nonorthogonal expansions
J.M.P. 27, 1271 (1986).

DAUBECHIES I., KLAUDER J.R., PAUL T.
Wiener measures for path integrals with affine kinematic variables.
J.M.P. 28, 85 (1987).

DAUBECHIES I., PAUL T.
Wavelets and applications.
in "VIIIth International Congress on Mathematical Physics" Proceedings -
Marseille, July 16-25, 1986 - M. Mebkhout, R. Seneor eds., p. 675.
World Scientific 1987.

DAUBECHIES I.
Frames of coherent states. Phase space localization and signal analysis.
Soumis aux Comm. in Pure and Applied Physics.

GRAFFI S., PAUL T.
Quantum avoided crossings for hydrogen atom in microwave electric field.
Soumis à Phys. Rev. Letters.

GROSSMANN A., MORLET J., PAUL T.
Transforms associated to square integrable group representations. I. General results.
J.M.P. 26, 2473 (1985).

GROSSMANN A., MORLET J., PAUL T.
Transforms associated to square integrable group representations. II. Examples.
Annales I.H.P. 45, 293 (1986).

GROSSMANN A.
Wavelet transforms and Edge detection.
in "Stochastic processes in physics and engineering",
Ph. Blanchard, L. Streit, M. Hazewinkel eds.

GROSSMANN A., HOLSCHNEIDER M., KRONLAND-MARTINET R., MORLET J.
Detection of abrupt changes in sound signals with the help of wavelet transforms.
Advances in Electronics and Electron. Phys.

HOLSCHNEIDER M.
On the wavelet transformation of fractal objects.
soumis à J. Stat. Phys.

JAFFARD S., MEYER Y.
Bases d'ondelettes dans des ouverts de \mathbb{R}^n .
A paraître.

KRONLAND-MARTINET R., MORLET J., GROSSMANN A.
Analysis of sound patterns through wavelet transforms.
Int. J. Pattern Recognition and Artificial Intelligence.

LEMARIE P.G.
Ondelettes à localisation exponentielle.
A paraître au Journal de Mathématiques Pures et Appliquées.

LEMARIE P.G., MEYER Y.
Ondelettes et bases hilbertiennes.
Revista Matematica Ibero Americana, vol. 2, 1 (1986).

MALLAT S.
A theory for multiscale signal decomposition : the scale change representation.
(GRAS Laboratory, Dept. of Computer and Information Science, Univ. of Pennsylvania,
Philadelphia, PA 19104-6389).

MEYER Y.
Wavelets and operator theory.
Cahiers des mathématiques de la décision.

MEYER Y.
Ondelettes, fonctions splines et analyses graduées.
Cahiers des mathématiques de la décision.

PAUL T.
Affine coherent states and the radial Schrödinger equation. I. Radial harmonic oscillator
and hydrogen atom.
Soumis aux Annales de l'I.H.P.

PAUL T.
Affine coherent states and the radial Schrödinger equation. II. Large N expansions.
Soumis aux Annales de l'I.H.P.

PAUL T.
Affine coherent states and the radial Schrödinger equation. III. Affine Wigner functions.
Soumis aux Annales de l'I.H.P.

PAUL T.
Ondelettes et mécanique quantique.
Thèse d'Etat - 19.12.1985 - Université d'Aix-Marseille II.

PAUL T.
A characterization of dilation analytic operators.
in "Schrödinger Operators, Aarhus 1985".
Lecture Notes in Maths. 1218 - Springer-Verlag.

TCHAMITCHIAN Ph.
Calcul symbolique sur les opérateurs de Calderon-Zygmund et bases inconditionnelles de
 $L^2(\mathbb{R})$.
C.R. Acad. Sc. Paris 303, Série 1, 215 (1986).

TCHAMITCHIAN Ph.
Bi-orthogonalité et théorie des opérateurs.
Soumis à Revista Ibero Americana.

LABORATOIRE DE MECANIQUE ET D'ACOUSTIQUE
31, Ch. Joseph-Aiguier (9e arr.) B.P. 71 - 13277 Marseille Cedex 8
TEL 91224000. TELEX CNRSMAR 430225F

